
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2016-06_D17wk23’.

2. Corrections required

2.1 Change 1 – Correction to function ‘f_TC_22_4_7_NBIOT()’

	Function name
	f_TC_22_4_7_NBIOT()

	Reason for change
	1. There is conflict between timer uplink data delay to delay the transmission of uplink user data and checking the UE should loop back data before timer uplink data delay expired at step 2A/5. And as Test Purpose 1 described, extended wait timer should be checked at step 2A. Change check time range to be within 60s (extendedwaitTime) at step 2A according to the associated Prose CR.
2. Considering above change 1, timer uplink data delay = 8s may be too short when RRC connnection release procedure takes 10s according to 36.331 subclause 5.3.8.3 without ack of release message. Change timer Uplink data delay of CLOSE UE TEST LOOP to 15s according to the associated Prose CR.
3. Considering above change 1 and check timer Uplink data delay of CLOSE UE TEST LOOP is unnecessary, delete all check about timer uplink data delay and superfluous steps 0D/3F.

	Summary of change
	1. Change check time range to be within 60s (extendedwaitTime) and use TimerToleranceMax of extendedwaitTime at step 2A.

2. Change v_DelayLoopbackDataTime to 15s instead of 8s according to the associated Prose CR.
3. Void superfluous steps 0D/3F, change f_NBIOT_RRC_ConnectionRequest_Def instead of f_NBIOT_RRC_ConnectionRequest_Common at step 5.

Associated Prose CR will be submitted in next RAN5 meeting.

	TTCN module
	NBIOT_RRC_ControlPlane.ttcn

	MCC160 Comment
	


Before change

  function f_TC_22_4_7_NBIOT() runs on NBIOT_PTC

  {//NB-IoT / RRC connection release with extendedWait / extendedWait ignored / RRC connection establishment / Reject with extendedWait

    var integer v_XtdWaitTime := 60;

    var integer v_DelayLoopbackDataTime := 8;

    var float v_WaitForLoopbackData := f_NBIOT_SetTimerToleranceMax (rrcTimer, int2float(v_DelayLoopbackDataTime));

    var float v_WaitForConnection := f_NBIOT_SetTimerToleranceMin (rrcTimer, int2float(v_XtdWaitTime));

    var NBIOT_CellInfo_Type v_CellInfo;

    var octetstring v_UserData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());

    var boolean v_TimerMsg4IsRunning := true;

    timer t_Wait;

    timer t_WaitForConnection;

    timer t_WaitForConfigMsg4;

    f_NBIOT_Init(c1);

    v_CellInfo := f_NBIOT_CellInfo_Get(nbiot_Cell1);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE2A_NB_TESTLOOP_ModeG);

    f_NBIOT_CloseUE_TestLoopModeGH(nbiot_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 1, 8);

    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 0" siclog@

    //@sic R5-172869 sic@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

    //@siclog "Step 0A" siclog@

    //@sic R5-171547 sic@

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3 are performed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 0B" siclog@

    //@sic R5-171547 sic@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData)))));//@sic R5s170364 sic@

    //@siclog "Step 0C" siclog@

    //@sic R5-171547 sic@

    //Wait for 1 s after the IP packet has been transmitted in step 0B.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    //@siclog "Step 0D" siclog@

    //@sic R5-172869 sic@

    //The SS starts timer Uplink data delay

    t_Wait.start(v_WaitForLoopbackData);
    //@siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionRelease-NB message on Ncell 1 with IE extendedwaitTime set to 60s.

    f_NBIOT_RRC_ConnectionRelease_Common(nbiot_Cell1, CONTROL_PLANE, cs_RRCConnectionRelease_NB(tsc_RRC_TI_Def, other, -, v_XtdWaitTime), cs_TimingInfo_Now);

    //@siclog "Step 2" siclog@

    //@sic R5-171547 sic@

    //Void.

    //@siclog "Step 2A" siclog@

    //@sic R5-172869 sic@

    //Check: Does the UE ignore extendedwaitTime and transmit an RRCConnectionRequest-NB message on Ncell 1 within 8s (Uplink data delay expiry) from step 0D?

    alt {

      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(mo_Data)))){

         t_Wait.stop;

         f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 2A");

        }

      [] t_Wait.timeout {

         f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 2A");

        }

    }

    //@siclog "Step 3" siclog@

    //@sic R5-171547 sic@

    //Void.

    //@siclog "Step 3A" siclog@

    //@sic R5-172869 sic@

    //Void.

    //@siclog "Step 3A1 - 3A3" siclog@

    //@sic R5-171547, R5-172869 sic@

    //Steps 2 to 4 of the 'Generic Test Procedure NB-IoT Control Plan CIoT MO user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.3.3  are performed

    //NOTE: The UE will transmit one ESM DATA TRANSPORT message containing loopback data received in step 0B.

    f_NBIOT_508Check_CP_MONonSMSDataTransfer(nbiot_Cell1, cr_UserDataContainer);

    //@siclog "Step 3A4" siclog@

    //@sic R5-171547, R5-172869 sic@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

    //@siclog "Step 3B" siclog@

    //@sic R5-171547 sic@

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3  are performed.

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 3C" siclog@

    //@sic R5-171547 sic@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData)))));//@sic R5s170364 sic@

    //@siclog "Step 3D" siclog@

    //@sic R5-171547 sic@

    //Wait for 1 s after the IP packet has been transmitted.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    //@siclog "Step 3E" siclog@

    //@sic R5-171547 sic@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

    //Configure Msg4 with RRCConnectionReject for step 6

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, cs_RRCConnectionReject_NB(v_XtdWaitTime));

    //@siclog "Step 3F" siclog@

    //@sic R5-172869 sic@

    //The SS starts timer Uplink data delay

    t_Wait.start(v_WaitForLoopbackData);
    //@siclog "Step 4" siclog@

    //@sic R5-171547 sic@

    //Void.

    //@siclog "Step 5" siclog@

    //UE transmits an RRCConnectionRequest-NB message within 8s (Uplink data delay expiry) from step 3E

    f_NBIOT_RRC_ConnectionRequest_Common(nbiot_Cell1, t_Wait);//@sic R5-172869 sic@

    //@siclog "Step 6" siclog@

    //The SS responds with RRCConnectionReject-NB message with IE extendedwaitTime set to 60s.

    //Automatically sent according to SS configuration

    //@siclog "Step 6A" siclog@

    //@sic R5-171547 sic@

    //Check: Does the UE transmit an RRCConnectionRequest-NB message on Ncell 1 within 60s?

    t_WaitForConfigMsg4.start(40.0);//start timer to configure Msg4 back to the default RRC Conn Setup in advance

    t_WaitForConnection.start(v_WaitForConnection);

    alt {

      // receive RRC Connection Request on SRB0 (CCCH):

      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(cr_EstablishmentCause_NB_Any)))

        {

          if(v_TimerMsg4IsRunning)

          {

            t_WaitForConfigMsg4.stop;

          }

          t_WaitForConnection.stop;

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 6A");

        }

      [] t_WaitForConnection.timeout

        {

        }

      [v_TimerMsg4IsRunning] t_WaitForConfigMsg4.timeout

        {

          //Now configure Msg4 back to the default RRC Conn Setup for the rest steps and postamble

          f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, f_NBIOT_508_RRCConnectionSetup());

          v_TimerMsg4IsRunning := false;

          repeat;

        }

    }

    //@siclog "Step 6B" siclog@

    //@sic R5-171547 sic@

    //The SS starts timer Timer_1 = 8 s

    t_Wait.start(v_WaitForLoopbackData);

    //EXCEPTION: Steps 6Ba1 to 6Bb1 describe a transaction that depends on the UE behaviour; the "lower case letter" identifies a test sequence that takes place if a specific behaviour happens.

    //Note 2: A UE may send the pending loopback data

    //@sic R5-171547 sic@

    alt {

      //@siclog "Step 6Ba1 - 6Ba4" siclog@

      //Steps 1 to 4 of the 'Generic Test Procedure NB-IoT Control Plan CIoT MO user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.3.3  are performed.

      //NOTE: The UE will transmit one ESM DATA TRANSPORT message containing loopback data received in step 3C.

      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(mo_Data)))){

        t_Wait.stop;

        f_NBIOT_508Check_CP_MONonSMSDataTransfer(nbiot_Cell1, cr_UserDataContainer);

        //@siclog "Step 6Ba5" siclog@

        //The SS transmits an RRCConnectionRelease-NB message

        f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

      }

      //@siclog "Step 6Bb1" siclog@

      //The SS waits for Timer_1 expiry

      [] t_Wait.timeout{}

    }

    //@siclog "Step 7 - 12b1" siclog@

    //@sic R5-171547 sic@

    //Void.

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble ( nbiot_Cell1, CONTROL_PLANE, N1_IDLE );

  }
After change

  function f_TC_22_4_7_NBIOT() runs on NBIOT_PTC

  {//NB-IoT / RRC connection release with extendedWait / extendedWait ignored / RRC connection establishment / Reject with extendedWait

    var integer v_XtdWaitTime := 60;

    var integer v_DelayLoopbackDataTime := 15;

    var float v_WaitForConnectionMax := f_NBIOT_SetTimerToleranceMax (rrcTimer, int2float(v_XtdWaitTime));

    var float v_WaitForConnection := f_NBIOT_SetTimerToleranceMin (rrcTimer, int2float(v_XtdWaitTime));

    var NBIOT_CellInfo_Type v_CellInfo;

    var octetstring v_UserData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());

    var boolean v_TimerMsg4IsRunning := true;

    timer t_Wait;

    timer t_WaitForConnection;

    timer t_WaitForConfigMsg4;

    f_NBIOT_Init(c1);

    v_CellInfo := f_NBIOT_CellInfo_Get(nbiot_Cell1);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE2A_NB_TESTLOOP_ModeG);

    f_NBIOT_CloseUE_TestLoopModeGH(nbiot_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 1, v_DelayLoopbackDataTime);
    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 0" siclog@

    //@sic R5-172869 sic@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

    //@siclog "Step 0A" siclog@

    //@sic R5-171547 sic@

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3 are performed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 0B" siclog@

    //@sic R5-171547 sic@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData)))));//@sic R5s170364 sic@

    //@siclog "Step 0C" siclog@

    //@sic R5-171547 sic@

    //Wait for 1 s after the IP packet has been transmitted in step 0B.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    //@siclog "Step 0D" siclog@

    //@sic R5-172869 sic@


 //Void.
    //@siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionRelease-NB message on Ncell 1 with IE extendedwaitTime set to 60s.

    f_NBIOT_RRC_ConnectionRelease_Common(nbiot_Cell1, CONTROL_PLANE, cs_RRCConnectionRelease_NB(tsc_RRC_TI_Def, other, -, v_XtdWaitTime), cs_TimingInfo_Now);

    //@siclog "Step 2" siclog@

    //@sic R5-171547 sic@

    //Void.

    //@siclog "Step 2A" siclog@

    //@sic R5-172869 sic@

    //Check: Does the UE ignore extendedwaitTime and transmit an RRCConnectionRequest-NB message on Ncell 1 within 8s (Uplink data delay expiry) from step 0D?

    t_WaitForConnection.start(v_WaitForConnectionMax);
    alt {

      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(mo_Data)))){

         t_WaitForConnection.stop;

         f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 2A");

        }

      [] t_WaitForConnection.timeout {

         f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 2A");

        }

    }

    //@siclog "Step 3" siclog@

    //@sic R5-171547 sic@

    //Void.

    //@siclog "Step 3A" siclog@

    //@sic R5-172869 sic@

    //Void.

    //@siclog "Step 3A1 - 3A3" siclog@

    //@sic R5-171547, R5-172869 sic@

    //Steps 2 to 4 of the 'Generic Test Procedure NB-IoT Control Plan CIoT MO user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.3.3  are performed

    //NOTE: The UE will transmit one ESM DATA TRANSPORT message containing loopback data received in step 0B.

    f_NBIOT_508Check_CP_MONonSMSDataTransfer(nbiot_Cell1, cr_UserDataContainer);

    //@siclog "Step 3A4" siclog@

    //@sic R5-171547, R5-172869 sic@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

    //@siclog "Step 3B" siclog@

    //@sic R5-171547 sic@

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3  are performed.

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 3C" siclog@

    //@sic R5-171547 sic@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData)))));//@sic R5s170364 sic@

    //@siclog "Step 3D" siclog@

    //@sic R5-171547 sic@

    //Wait for 1 s after the IP packet has been transmitted.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    //@siclog "Step 3E" siclog@

    //@sic R5-171547 sic@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

    //Configure Msg4 with RRCConnectionReject for step 6

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, cs_RRCConnectionReject_NB(v_XtdWaitTime));

    //@siclog "Step 3F" siclog@

    //@sic R5-172869 sic@


 //Void.
    //@siclog "Step 4" siclog@

    //@sic R5-171547 sic@

    //Void.

    //@siclog "Step 5" siclog@

    //UE transmits an RRCConnectionRequest-NB message within 8s (Uplink data delay expiry) from step 3E

 f_NBIOT_RRC_ConnectionRequest_Def(nbiot_Cell1);
    //@siclog "Step 6" siclog@

    //The SS responds with RRCConnectionReject-NB message with IE extendedwaitTime set to 60s.

    //Automatically sent according to SS configuration

    //@siclog "Step 6A" siclog@

    //@sic R5-171547 sic@

    //Check: Does the UE transmit an RRCConnectionRequest-NB message on Ncell 1 within 60s?

    t_WaitForConfigMsg4.start(40.0);//start timer to configure Msg4 back to the default RRC Conn Setup in advance

    t_WaitForConnection.start(v_WaitForConnection);

    alt {

      // receive RRC Connection Request on SRB0 (CCCH):

      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(cr_EstablishmentCause_NB_Any)))

        {

          if(v_TimerMsg4IsRunning)

          {

            t_WaitForConfigMsg4.stop;

          }

          t_WaitForConnection.stop;

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 6A");

        }

      [] t_WaitForConnection.timeout

        {

        }

      [v_TimerMsg4IsRunning] t_WaitForConfigMsg4.timeout

        {

          //Now configure Msg4 back to the default RRC Conn Setup for the rest steps and postamble

          f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, f_NBIOT_508_RRCConnectionSetup());

          v_TimerMsg4IsRunning := false;

          repeat;

        }

    }

    //@siclog "Step 6B" siclog@

    //@sic R5-171547 sic@

    //The SS starts timer Timer_1 = 8 s

    t_Wait.start(8.0);

    //EXCEPTION: Steps 6Ba1 to 6Bb1 describe a transaction that depends on the UE behaviour; the "lower case letter" identifies a test sequence that takes place if a specific behaviour happens.

    //Note 2: A UE may send the pending loopback data

    //@sic R5-171547 sic@

    alt {

      //@siclog "Step 6Ba1 - 6Ba4" siclog@

      //Steps 1 to 4 of the 'Generic Test Procedure NB-IoT Control Plan CIoT MO user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.3.3  are performed.

      //NOTE: The UE will transmit one ESM DATA TRANSPORT message containing loopback data received in step 3C.

      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(mo_Data)))){

        t_Wait.stop;

        f_NBIOT_508Check_CP_MONonSMSDataTransfer(nbiot_Cell1, cr_UserDataContainer);

        //@siclog "Step 6Ba5" siclog@

        //The SS transmits an RRCConnectionRelease-NB message

        f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

      }

      //@siclog "Step 6Bb1" siclog@

      //The SS waits for Timer_1 expiry

      [] t_Wait.timeout{}

    }

    //@siclog "Step 7 - 12b1" siclog@

    //@sic R5-171547 sic@

    //Void.

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble ( nbiot_Cell1, CONTROL_PLANE, N1_IDLE );

  }
